# $Id$
# Contributor: damir <damir@archlinux.org>
# Contributor: Daniel J Griffiths <ghost1227@archlinux.us>
# Maintainer: Gaetan Bisson <bisson@archlinux.org>

pkgname=anthy
pkgver=9100h
_filecode=37536
pkgrel=4
pkgdesc='Hiragana text to Kana Kanji mixed text Japanese input method'
arch=('i686' 'x86_64')
url='http://sourceforge.jp/projects/anthy/'
license=('LGPL' 'GPL')
source=("http://dl.sourceforge.jp/anthy/${_filecode}/${pkgname}-${pkgver}.tar.gz")
sha256sums=('d256f075f018b4a3cb0d165ed6151fda4ba7db1621727e0eb54569b6e2275547')

build() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	./configure --prefix=/usr --sysconfdir=/etc
	make
}

package() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	make EMACS=emacs DESTDIR="${pkgdir}" install
}
